


HEWLETT-PACKARD

HP-41C

MANUEL
D'APPLICATIONS





HP-41C

Manuel d'applications

Juillet 1979

00041-90028

© 1979, Hewlett-Packard France.

Texte protégé par la législation en vigueur en matière de propriété littéraire et dans tous les pays.

Imprimé aux U.S.A.

Table des matières

Introduction	5
Les tableaux d'utilisation	7
Chargement d'un programme	9
Notation polonaise inverse	11
Calculs de dates	17
Jeux	21
La leçon d'arithmétique	25
Conversion hexadécimal-décimal	31
Calculs financiers	35
Recherche de racines	41
Ajustement de courbes	45
Calcul vectoriel	53
Jeu du Blackjack	57

Les programmes figurant dans ce fascicule sont sans garantie d'aucune sorte. Par conséquent, la société Hewlett-Packard n'assume aucune responsabilité consécutive ou non à l'utilisation de ces programmes ou documents.

Introduction

Ce manuel présente tout un ensemble de programmes d'applications qui illustrent la puissance et la souplesse de programmation du HP-41C. Ces programmes ont été conçus pour être à la fois utiles, enrichissants et divertissants. La mémoire permanente vous permet de les conserver en mémoire, prêts à être utilisés. Chacun de ces dix programmes est présenté avec un tableau d'utilisation, un ou plusieurs exemples et des éclaircissements sur certaines techniques de programmation «Astuces de programmation».

L'étude de ces programmes préparés par des professionnels vous aidera à étendre vos connaissances en programmation. La possibilité de personnaliser votre calculateur vous permet d'imaginer de nouvelles applications auxquelles vous n'aviez pas songé.

Avant d'utiliser l'un de ces programmes, lisez attentivement le début de ce manuel consacré aux tableaux d'utilisation et chargement d'un programme. Mais avant tout, nous vous conseillons de lire le «*Manuel d'utilisation et guide de programmation du HP-41C*».

Pour exécuter un programme, chargez-le dans le calculateur en suivant pas à pas le listage donné pour ce programme puis reportez-vous au tableau d'utilisation pour la procédure à suivre. Celle-ci est très simple à assimiler et vous prendrez rapidement l'habitude de répondre directement aux messages donnés à l'affichage par l'introduction des données ou des séquences de touches requises.

Les «astuces de programmation»: ce sont des techniques de programmation intéressantes, utilisées dans ces programmes d'applications, et que vous pouvez utilement exploiter dans vos propres programmes. Par ailleurs, nous vous conseillons de lire attentivement les commentaires des listages de programmes: ils constituent un excellent enseignement.

A l'exception du programme 10 (jeu du Blackjack), vous pouvez enregistrer tous les programmes présentés dans ce manuel sur la version de base du HP-41C. Le Blackjack nécessite la connexion d'un module mémoire supplémentaire. Avec une configuration mémoire supérieure, le HP-41C vous permettra d'utiliser ces programmes à l'intérieur de certains de vos programmes plus importants.

Vous pourrez même les modifier à votre convenance. Ainsi, la programmation peut devenir un jeu passionnant...

Les tableaux d'utilisation

Un tableau d'utilisation accompagne chaque programme de cette bibliothèque: il vous indique la procédure à suivre pour utiliser au mieux vos programmes.

Il se présente sous la forme de 5 colonnes intitulées: *N°*, *Instructions*, *Données*, *Fonctions*, *Affichage*.

- *N°*: donne les numéros des instructions utilisateur (pas de procédure);
- *Instructions*: indique les opérations à effectuer;
- *Données*: indique les valeurs à introduire et éventuellement l'unité ou la réponse alphanumérique correcte à la question posée. Les touches utilisées sont: les touches numériques de 0 à 9, le point décimal, **[EEX]** (exposant) et **[CHS]** (changement de signe).
- *Fonctions*: indique la ou les touches à utiliser après l'introduction des données. Lorsqu'une information apparaît en jaune dans la colonne Données ou Fonctions, appuyez sur la touche **[ALPHA]**, mode alphanumérique.
Par exemple **[XEQ] A4C** signifie: appuyez sur **[XEQ] [ALPHA] A ■ 4C [ALPHA]**. Vous pouvez aussi affecter la fonction A4C à une touche, en effectuant: **■ [ASN] [ALPHA] A4C [ALPHA]** touche. Il suffit alors d'appuyer sur cette touche en mode personnel pour exécuter la fonction.
- *Affichage*: donne l'ensemble des messages numériques et alphanumériques affichés au cours du programme.

Au-dessus de la colonne Affichage, une case étiquetée Taille indique le nombre minimum de registres nécessaires à l'exécution du programme. Nous vous conseillons d'affecter une taille suffisante (instruction SIZE) à la mémoire programme avant de charger un programme. Consultez le manuel d'utilisation du HP-41C qui donne des précisions sur cette opération.

Chargement d'un programme

Nous vous donnons ici les points principaux à retenir lors du chargement d'un programme. Les listages figurant dans ce manuel ont été faits sur une imprimante HP 82143. Ils vous permettent de comprendre la procédure d'introduction des programmes en l'absence de toute indication de pressions de touches.

Voici cette procédure :

1. A la fin de chaque listage de programme se trouve un listage de l'état du calculateur pour une exécution correcte du programme (en particulier l'emplacement mémoire mobilisé-SIZE). Avant le chargement, appuyez sur **XEQ** **ALPHA** SIZE **ALPHA** et spécifiez la taille mémoire désirée (par trois chiffres – exemple : 010). (Pour les programmes des HP 67/HP 97, il faut spécifier 026.)

Ce listage d'état donne aussi le format d'affichage et l'état des indicateurs binaires utilisés par le programme. Vérifiez donc si le format d'affichage du HP-41C et les indicateurs binaires sont conformes au listage :

2. Mettez le calculateur en mode programme (en appuyant sur la touche **PRGM**) et appuyez sur **GTO** **▣** **▣**. Le calculateur est prêt.
3. Introduisez le programme à partir de son listage. Quelques remarques :
 - a) Si des guillemets (") entourent un caractère ou un groupe de caractères, il s'agit là de texte qu'il faut introduire en mode ALPHA : appuyez sur **ALPHA**, introduisez ces caractères et appuyez à nouveau sur **ALPHA**. Par exemple, 06 "EXEMPLE" est introduit de la manière suivante : **ALPHA** EXEMPLE **ALPHA**. Le caractère blanc est introduit par la touche **SPACE**.
 - b) Le losange figurant devant chaque instruction LBL est destiné uniquement à améliorer la lisibilité du listage. Vous l'ignorez lors du chargement du programme.
 - c) Le symbole de la division est / dans le listage de l'imprimante. Pour l'introduire, appuyez sur **÷**.
 - d) Le symbole de la multiplication est * dans le listage de l'imprimante. Pour l'introduire, appuyez sur **×**.
 - e) Le caractère † figurant sur certaines lignes du listage indique une fonction **APPEND**. Lorsque vous le rencontrez, appuyez sur **APPEND** en mode ALPHA. Par exemple, 03 "† EXEMPLE" est introduit de la manière suivante **ALPHA** **APPEND** EXEMPLE **ALPHA**.

10 Chargement d'un programme


- f) Les adresses doivent prendre une des formes suivantes:
- nn (nombre à deux chiffres);
 - IND nn (adressage indirect: ■, suivie par un nombre à deux chiffres);
 - X, Y, Z, T ou L (adressage de la pile: □ suivie de X, Y, Z, T ou L);
 - IND X, Y, Z, T ou L (adressage indirect de la pile ■ □ suivies de X, Y, Z, T ou L).

Listage	Séquence de touches	Affichage
01 *LBL "SAMPLE"	■ [LBL] [ALPHA] SAMPLE [ALPHA]	01 LBLTSAMPLE
02 "THIS IS A"	[ALPHA] THIS IS A [ALPHA]	02THIS IS A
03 "FSAMPLE"	[ALPHA] ■ [APPEND] SAMPLE	03TSAMPLE
04 AVIEW	■ AVIEW [ALPHA]	04 AVIEW
05 6	6	05 6
06 ENTER↑	[ENTER↑]	06 ENTER ↗
07 -2	2 [CHS]	07 -2
08 /	[÷]	08 /
09 ABS	[XEQ] [ALPHA] ABS [ALPHA]	09 ABS
10 STO IND L	[STO] ■ □ L	10 STO IND L
11 R3="	[ALPHA] R3= ■ [ARCL] 03	11 R3=
12 ARCL 03	■ [AVIEW]	12 ARCL 03
13 AVIEW	[ALPHA]	13 AVIEW
14 RTN	■ [RTN]	14 RTN

Notation polonaise inverse

Ce programme est conçu pour vous aider à comprendre et à utiliser la notation polonaise inverse. Les quatre registres de la pile opérationnelle sont visibles simultanément : vous pouvez ainsi voir, plutôt que supposer, l'effet d'une séquence de pressions de touches. Les fonctions fournies sont montrées sur le clavier ci-dessous. Il est possible d'observer l'effet sur la pile, de fonctions qui ne sont pas incluses dans ce programme. Exécutez simplement la fonction désirée, puis appuyez sur la touche **R/S** à laquelle la fonction STACK est affectée. Les seules différences de fonctionnement entre cette machine fictive et la vraie, sont :

- vous ne devez introduire que des nombres d'un chiffre ;
- les fonctions STO et RCL adressent un seul registre.



12 Notation polonaise inverse

				Taille: 001
N°	Instructions	Données	Fonctions	Affichage
1	Introduisez le programme.			
2	Affectez ces routines et passez en mode personnel. Les affectations suggérées sont celles du clavier de la page précédente. SWAP $\boxed{x^2y}$ RDWN $\boxed{R\downarrow}$ ST \boxed{STO} RC \boxed{RCL} E \uparrow $\boxed{ENTER\uparrow}$ CHSN \boxed{CHS} CLR $\boxed{\leftarrow}$ MI $\boxed{-}$ PL $\boxed{+}$ DI $\boxed{\div}$ MU $\boxed{\times}$ DI $\boxed{-}$ 9 9 8 8 7 7 6 6 5 5 4 4 3 3 2 2 1 1 0 0 LSTX \boxed{LASTx} STACK $\boxed{R/S}$			
3	Appuyez sur les touches désirées et regardez les contenus de la pile en train de changer.			
4	Les fonctions $\boxed{R\uparrow}$ et \boxed{CLST} K sont obtenues par et Pour affecter une fonction, par exemple FCN à la touche $\boxed{\sqrt{x}}$, exécutez:		\boxed{XEQ} RUP \boxed{XEQ} CLSTK $\boxed{\square}$ ASN \boxed{ALPHA} FCN \boxed{ALPHA} $\boxed{\sqrt{x}}$	

Exemple 1:

Évaluez l'expression

$$\frac{(2 + b)b}{8 - b}$$

pour $b = 3$

Appuyez sur:

Fonction	Affichage
\boxed{XEQ} \boxed{ALPHA} CLSTK \boxed{ALPHA}	X:0 Y:0 Z:0 T:0
2	X:2 Y:0 Z:0 T:0
$\boxed{ENTER\uparrow}$	X:2 Y:2 Z:0 T:0
3	X:3 Y:2 Z:0 T:0
$\boxed{+}$	X:5 Y:0 Z:0 T:0
$\boxed{\square}$ \boxed{LASTx}	X:3 Y:5 Z:0 T:0
$\boxed{\times}$	X:15 Y:0 Z:0 T:0
8	X:8 T:15 Z:0 T:0
$\boxed{\square}$ \boxed{LASTx}	X:3 Y:8 Z:15 T:0
$\boxed{-}$	X:5 Y:15 Z:0 T:0
$\boxed{+}$	X:3 Y:0 Z:0 T:0

Exemple 2:

Sans modifier les résultats ci-dessus, calculez

$$\frac{2 + 4(9 - 7)}{6 - 4}$$

Fonction	Affichage
9	X:9 Y:3 Z:0 T:0
ENTER	X:9 Y:9 Z:3 T:0
7	X:7 Y:9 Z:3 T:0
-	X:2 Y:3 Z:0 T:0
4	X:4 Y:2 Z:3 T:0
X	X:8 Y:3 Z:0 T:0
2	X:2 Y:8 Z:3 T:0
+	X:10 Y:3 Z:0 T:0
6	X:6 Y:10 Z:3 T:0
ENTER	X:6 Y:6 Z:10 T:3
4	X:4 Y:6 Z:10 T:3
-	X:2 Y:10 Z:3 T:3
+	X:5 Y:3 Z:3 T:3

Après un **ENTER** la pile ne monte pas lorsqu'une donnée est introduite.

Remarquez que le résultat (3) de l'exemple 1 ne perturbe pas l'exécution de l'exemple 2.

Exemple 3:

Convertissez le nombre complexe $3 + 4i$ dans sa forme polaire.

Fonction	Affichage
4	X:4 Y:5 Z:3 T:3
ENTER	X:4 Y:4 Z:5 T:3
3	X:3 Y:4 Z:5 T:3
R-P	5
STACK	X:5 Y:53 Z:5 T:3

Souvenez-vous que STACK est affectée à **R/S**.

Astuce de programmation

Une des caractéristiques intéressantes de ce programme est la routine d'affichage de la pile: STACK. Vous pouvez ainsi à tout moment vérifier les contenus de la pile au cours du traitement de vos propres problèmes.

14 Notation polonaise inverse

<pre> 01+LBL "CLS TK" 02 CLST 03 GTO 14 04+LBL "1" 05 FS?C 05 06 CLX 07 1 08 GTO 14 09+LBL "2" 10 FS?C 05 11 CLX 12 2 13 GTO 14 14+LBL "3" 15 FS?C 05 16 CLX 17 3 18 GTO 14 19+LBL "4" 20 FS?C 05 21 CLX 22 4 23 GTO 14 24+LBL "5" 25 FS?C 05 26 CLX 27 5 28 GTO 14 29+LBL "6" 30 FS?C 05 31 CLX 32 6 33 GTO 14 34+LBL "7" 35 FS?C 05 36 CLX 37 7 38 GTO 14 39+LBL "8" 40 FS?C 05 41 CLX 42 8 43 GTO 14 44+LBL "9" 45 FS?C 05 46 CLX 47 9 48 GTO 14 49+LBL "0" </pre>	<p>Effacement de la pile.</p> <p>Si le décalage pile est invalidé effacement de x.</p> <p>Introduction de 1.</p> <p>Introduction de 2.</p> <p>Introduction de 3.</p> <p>Introduction de 4.</p> <p>Introduction de 5.</p> <p>Introduction de 6.</p> <p>Introduction de 7.</p> <p>Introduction de 8.</p> <p>Introduction de 9.</p>	<pre> 50 FS?C 05 51 CLX 52 0 53 GTO 14 54+LBL 13 55 CF 05 56+LBL 14 57+LBL "STA CK" 58 "X:" 59 ARCL X 60 "F Y:" 61 ARCL Y 62 "F Z:" 63 ARCL Z 64 "F T:" 65 ARCL T 66 AVIEW 67 RTN 68+LBL "E↑" 69 SF 05 70 ENTER↑ 71 GTO 14 72+LBL "RDW N" 73 RDN 74 GTO 13 75+LBL "SWA P" 76 X<>Y 77 GTO 14 78+LBL "RUP " 79 R↑ 80 GTO 13 81+LBL "PL" 82 + 83 GTO 13 84+LBL "MI" 85 - 86 GTO 13 87+LBL "MU" 88 * 89 GTO 13 90+LBL "DI" 91 / 92 GTO 13 93+LBL "CLR " 94 SF 05 </pre>	<p>Introduction de 0.</p> <p>Validation du décalage pile.</p> <p>Affichage de la pile.</p> <p>Invalidation du décalage pile.</p> <p>Permutation vers le bas.</p> <p>Echange de x et y.</p> <p>Permutation vers le haut.</p> <p>Plus.</p> <p>Moins.</p> <p>Multiplié.</p> <p>Divisé.</p>
<p>R00 Stockage</p>			

<p>95 CLX 96 GTD 14 97+LBL "CHS N" 98 CHS 99 GTD 14 100+LBL "ST" 101 STD 00 102 GTD 14 103+LBL "RC" 104 FS?C 05 105 CLX 106 RCL 00 107 GTD 14 108+LBL "LST X" 109 FS?C 05 110 CLX 111 LASTX 112 GTD 14</p> <p>Etat: Size=01 Fix 0</p> <p>Indicateurs utilisés F05 armé=mouvements de la pile interdits F29 désarmé=pas de point pour séparer les groupes de trois chiffres.</p>	<p>Invalidation du décalage de la pile. Effacement de x.</p> <p>Changement du signe.</p> <p>Stockage.</p> <p>Si décalage pile invalidé, effacement de x. Rappel.</p>		
--	--	--	--

Note: Il peut être judicieux d'affecter FS?C à une touche, par exemple à **[LN]**. Vous pouvez alors appuyer une 1^{re} fois sur **[LN]** pour afficher FS?C--, puis une seconde fois pour créer FS?C 05. Vous devez pour cela être en mode personnel.

Calculs de dates

Ce programme résout toutes sortes de problèmes de calendrier :

- Nombre de jours compris entre deux dates.
- Calcul d'une date à partir d'une autre date, connaissant un nombre de jours.
- Détermination du jour de la semaine pour une date donnée. Le jour est affiché en toutes lettres.

Les dates sont introduites sous la forme :

mm,jjaaaa. Elles sont affichées en :

mois jj,aaaa.

				Taille : 010
N°	Instructions	Données	Fonctions	Affichage
1	Chargez le programme. Passez en mode personnel. Calcul du jour de la semaine			
2a	Introduisez la date et calculez le jour.	DATE	E	Jour
3a	Recommencez l'opération 2a pour une nouvelle date. Calcul du nombre de jours entre deux dates			
2b	Introduisez deux des données suivantes: 1 ^{re} date 2 ^e date Nombre de jours entre deux dates.	D1 D2 D	A B C	Date 1 Date 2 D
3b	Effectuez l'un des calculs suivants: 1 ^{re} date 2 ^e date Nombre de jours entre deux dates.		A B C	Date 1 Date 2 D
4	Recommencez l'opération 2b pour de nouvelles valeurs (celles qui ne changent pas n'ont pas besoin d'être réintroduites).			

Exemple 1 :

Quel jour de la semaine fut le 19 février 1946?

Appuyez sur
2.191946 **E**

Affichage
TUESDAY

Exemple 2:

Quelle est la date correspondant à 10 000 jours après le 4 août 1978 ?

Appuyez sur 8.041978 <input type="checkbox"/> A 10000 <input type="checkbox"/> C <input type="checkbox"/> B	→	Affichage DEC 20.2005
---	---	---------------------------------

Exemple 3:

Un homme né le 18 décembre 1913, a un fils né le 19 février 1946. A quelle date le père a-t-il eu le double de l'âge du fils ?

Appuyez sur 12.181913 <input type="checkbox"/> A	→	Affichage DEC 18.1913	
2.191946 <input type="checkbox"/> B	→	FEB 19.1946	
<input type="checkbox"/> C	→	11751	
2 <input checked="" type="checkbox"/> <input type="checkbox"/> C <input type="checkbox"/> B	→	APR 23.1978	Nombre de jours. A cette date le père a eu le double de l'âge du fils.

Astuce de programmation

Ce programme utilise la caractéristique que possède le HP-41C d'établir une notation alphanumérique pour l'affichage de la date. A partir d'une date de la forme mm.jjaaaa dans le registre X, l'instruction EQ IND X exécute une sous-routine qui place le mois (en trois lettres) spécifié dans la registre ALPHA. Le programme multiplie ensuite par 100 la partie décimale du contenu de X, désarme l'indicateur de point décimal, et ajoute à l'affichage le jour et l'année ainsi constitués. Par exemple la valeur initiale: 12.251978 du registres X devient DEC 25, 1978 à l'affichage.

<pre> 103 ENTER↑ 104 XEQ IND X 105 FRC 106 1 E2 107 * 108 CF 28 109 FIX 4 110 ARCL X 111 RDN 112 AVIEW 113 SF 28 114 RTN 115♦LBL 22 116 INT 117 ST+ 09 118 12 119 * 120 - 121 RTN 122♦LBL C 123 CF 29 124 FIX 0 125 STO 01 126 FS?C 22 127 RTN 128 RCL 04 129 RCL 03 130 - 131 STO 01 132 RTN 133♦LBL E 134 SF 06 135 SF 22 136 RCL 05 137 5 138 XEQ 20 139 RCL IND 02 140 7 141 MOD 142 13 143 + 144 XEQ IND X 145 AVIEW 146 RTN 147♦LBL 13 148 "FRIDAY" 149 RTN 150♦LBL 14 151 "SATURDA Y" 152 RTN </pre>	<p>Jour de la semaine.</p>	<pre> 153♦LBL 15 154 "SUNDAY" 155 RTN 156♦LBL 16 157 "MONDAY" 158 RTN 159♦LBL 17 160 "TUESDAY" " 161 RTN 162♦LBL 18 163 "WEDNESD AY" 164 RTN 165♦LBL 19 166 "THURSDA Y" 167 RTN 168♦LBL 01 169 "JAN " 170 RTN 171♦LBL 02 172 "FEB " 173 RTN 174♦LBL 03 175 "MAR " 176 RTN 177♦LBL 04 178 "APR " 179 RTN 180♦LBL 05 181 "MAY " 182 RTN 183♦LBL 06 184 "JUN " 185 RTN 186♦LBL 07 187 "JUL " 188 RTN 189♦LBL 08 190 "AUG " 191 RTN 192♦LBL 09 193 "SEP " 194 RTN 195♦LBL 10 196 "OCT " 197 RTN 198♦LBL 11 199 "NOV " 200 RTN 201♦LBL 12 202 "DEC " </pre>	
<p>Etats Taille=010 Fix 4 Indicateurs utilisés F06 F28</p>			


Jeux

Ce programme est une version de jeu du «pendu» (avec des mots). Le joueur N° 1 enregistre dans le calculateur un mot de 6 lettres. Le joueur N° 2 essaie de deviner ce mot en proposant des lettres. Après chacune de ces propositions, le calculateur affiche les lettres justes. – Chaque caractère juste est écrit à sa place entre les crochets. Lorsque le mot est trouvé, le nombre de lettres proposées est affiché.

				Taille: 020
N°	Instructions	Données	Fonctions	Affichage
1	Introduisez le programme.			
2	Lancez l'exécution du programme.		[XEQ] WORDS	KEY IN WORD
3	Premier joueur: introduisez votre mot (tout mot de 6 caractères).	xxxxxx	[R/S]	LETTER?
4	Second joueur: essayez une lettre.	x	[R/S]	<lettres justes> LETTER?
5	Répétez l'opération 4 jusqu'à la fin du jeu; affichage de DONE: le HP-41C affiche ensuite WORD IS <mot> puis: YOU TOOK nn GUESSES.			

Exemple:

Introduisez «HP-41C», puis cherchez-le.



Notez que le programme s'arrête en mode ALPHA.

Astuce de programmation

Des routines du type de SPEL et DESPEL ont été utilisées pour le développement de ce programme: SPEL et DESPEL. Elles ont pour rôle respectif de construire un mot à partir d'un ensemble de lettres et de décomposer un mot en ses différentes lettres. Seule DESPEL figure dans le programme final parce que SPEL a déjà été exécutée dans la partie de programme effectuant la comparaison des lettres.

Un code doit être transféré aux deux routines SPEL et DESPEL, à travers le registre X. Ce code indique à SPEL où trouver ses lettres, et à DESPEL où les mettre. Il est de la forme:

pl. Odl pour SPEL

ou: dl. Opp pour DESPEL

où: dl = registre pour la dernière lettre

pl = registre pour la première lettre

pp = pl - 1

Des routines du type de SPEL et DESPEL permettent de coder et de décoder une grande variété de chaînes. A titre d'exercice vous pouvez vous amuser à rechercher comment DESPEL est utilisée dans le programme de conversion hexadécimale (page 26).

Voici ces deux routines:

01 *LBL "SPE	Suppose le registre	01 *LBL "DES	
L "	ALPHA effacé.	PEL "	
02 STO 07	Stocke le compteur	02 STO 07	Stocke le compteur
03 *LBL 08	pl. Odl.		dl. Opp.
04 ARCL IND		03 ASTO 00	Stocke le mot.
07		04 *LBL 07	
05 ISG 07	Construction du	05 " "	
06 GTO 08	mot: si cette lettre	06 ARCL 00	
07 RTN	n'est pas la	07 ASTO 00	Stocke toutes les
	dernière...		lettres sauf la
	répète la boucle.		dernière lettre.
		08 ASHF	Stocke la dernière
		09 ASTO IND	lettre.
		07	
		10 DSE 07	
		11 GTO 07	Reprendre la boucle
		12 RTN	s'il y a encore des
			lettre.

<pre> 01+LBL "WOR DS" 02 "KEY IN WORD" 03 AON 04 PROMPT 05 ASTO 08 06 6 07 XEQ "DES PEL" 08 .9 09 STO 17 10 " " 11 ASTO 09 12 16.01 13 XEQ "DES PEL" 14+LBL "LTT R" 15 CLA 16 ASTO 09 17 "LETTER?" " 18 AON 19 PROMPT 20 ASTO 10 21 ISG 17 22 1.006 23 STO 18 24+LBL 06 25 " " 26 ASTO Y 27 RCL 18 28 10 29 + 30 CLA 31 ARCL IND X 32 RDN 33 ASTO X 34 X=Y? 35 GTO 00 36 CLA 37 ARCL 10 38 ASTO Y 39 CLA 40 ARCL IND 18 41 ASTO X 42 X=Y? </pre>	<p>Stockage du mot à deviner. Ventilation des lettres dans les registres R01 à R06.</p> <p>Placement de blancs dans les registres R11 à R16.</p> <p>Demande d'une lettre au joueur.</p> <p>Stockage de la lettre. Comptage des lettres. Initialisation du compteur. Début de la boucle 6.</p> <p>Si cette position est déjà affectée d'une lettre, alors affichage.</p> <p>Si la lettre est juste.</p>	<pre> 43 GTO 00 44 " " 45 ASTO X 46+LBL 00 47 CLA 48 ARCL 09 49 ARCL X 50 ASTO 09 51 AVIEW 52 10 53 RCL 18 54 + 55 CLA 56 ARCL Y 57 ASTO IND X 58 ISG 18 59 GTO 06 60 CLA 61 ARCL 08 62 ASTO Y 63 CLA 64 ARCL 09 65 ASTO X 66 X=Y? 67 GTO 00 68 PSE 69 PSE 70 GTO "LTT R" 71+LBL 00 72 "DONE" 73 AVIEW 74 "WORD IS <" 75 ARCL 09 76 "F>" 77 AVIEW 78 PSE 79 PSE 80 RCL 17 81 INT 82 "YOU TOO K " 83 ARCL X 84 "F GUESS ES" 85 AVIEW 86 RTN 87+LBL "DES PEL" </pre>	<p>Alors affichage de i sinon, affichage blanc.</p> <p>Ajoute une lettre à l'affichage.</p> <p>Répétition de la boucle six fois.</p> <p>Si les mots de X et Y sont identiques le joueur a gagné. Sinon, demande d'une autre lettre.</p> <p>Affichage du mot.</p> <p>Affichage du nombre de tentatives.</p>
--	--	--	--

R00=Temporaire
R01=1^{re} lettre, SW
R02=2^e lettre, SW
R03=3^e lettre, SW
R04=4^e lettre, SW
R05=5^e lettre, SW
R06=6^e lettre, SW

R07=Compteur
R08=Mot secret (SW)
R09=Mot du joueur (PW)
R10=Lettre courante
R11=1^{re} lettre, PW
R12=2^e lettre, PW
R13=3^e lettre, PW

<pre> 88 STO 07 89 ASTO 00 90+LBL 07 91 " " 92 ARCL 00 93 ASTO 00 94 ASHF 95 ASTO IND 07 96 DSE 07 97 GTO 07 98 RTN Etat: Size=19 Fix 0 CF 29 F29 désarmé pour supprimer la virgule décimale. </pre>	<p>Sous-programme pour décomposer le mot en lettres.</p>		
<p>R14=4^e lettre, MJ R15=5^e lettre, MJ R16=6^e lettre, MJ R17=Compteur R18=Compteur</p>			

La leçon d'arithmétique

Ce programme génère une suite d'exercices destinés au calcul mental, de difficulté variable et comprenant: addition, soustraction, multiplication ou division. Après 10 problèmes, le calculateur affiche le pourcentage de réponses correctes (deux réponses sont autorisées pour chaque problème). Le programme est appelé par **[XEQ] [ALPHA] TEACH [ALPHA]**. Le calculateur demande alors la valeur maximale des nombres utilisés dans les différentes opérations. Vous l'introduisez, vous appuyez sur **[R/S]**: le calculateur affiche: +, -, *, /? (l'indicateur ALPHA est allumé). Appuyez ensuite sur la touche SHIFT, et sur l'une des fonctions arithmétiques, puis sur **[R/S]** pour commencer l'exercice.

Pour chacun des problèmes affichés, introduisez votre réponse et appuyez sur **[R/S]**. Si la réponse est correcte, un «OUI» est affiché et un nouveau problème apparaît. Sinon, un bruit désagréable accompagne le message «NON» et une seconde réponse vous est accordée. Si la réponse est à nouveau erronée, le calculateur affiche le résultat correct et pose le problème suivant. Lorsque les 10 problèmes sont effectués, le calculateur vous donne votre score (en pourcentage) et «joue un air de musique» si vous avez eu les 10 bonnes réponses du premier coup. Ensuite le programme recommence par l'affichage de la question «+, -, *, /?».

Les séries de problèmes sont déterminées par une racine prise dans l'intervalle 0–1 placée dans le registre X lorsque vous commencez le programme. Si vous voulez recommencer la même série de problèmes, il vous suffit d'introduire la même racine. Si vous n'introduisez pas de racine le calculateur utilise le nombre présent dans le registre X.

				Taille: 010
N°	Instructions	Données	Fonctions	Affichage
1	Introduisez le programme.			
2	Introduisez une racine ($0 \leq \text{racine} < 1$) et initialisez le programme.	Racine	[XEQ] TEACH	MAX NUMBER?
3	Introduisez la valeur maximale des nombres.	N	[R/S]	+, -, *, / ?
4	Choisissez le type d'opération: Addition Soustraction Multiplication Division	+ - * /	[R/S] [R/S] [R/S] [R/S]	n1 + n2 = ? n1 - n2 = ? n1 * n2 = ? n1 / n2 = ?
5	Introduisez votre réponse.	Réponse	[R/S]	YES ou NO
6	A la fin des 10 exercices, le pourcentage de bonnes réponses est affiché et vous pouvez recommencer une nouvelle série d'exercices à partir de l'opération 4.			SCORE % RIGHT

Messages de demande des équations

$$(n_1) + (n_2) = ?$$

$$(n_1) - (n_2) = ?$$

$$(n_1) * (n_2) = ?$$

$$(n_1) / (n_2) = ?$$

Exemple :

Avec la racine 0,021946, effectuez quelques problèmes de soustraction en utilisant des nombres inférieurs ou égaux à 14.

Appuyez sur	Affichage
.021946	
XEQ ALPHA TEACH ALPHA →	MAX NUMBER ?
14 R/S →	+, -, *, / ?
[-] R/S →	12 - 5 = ?
7 R/S →	YES
	14 - 13 = ?
1 R/S →	YES
	13 - 6 = ?
8 R/S →	NO 13 - 6 = ?
7 R/S →	YES
	14 - 11 = ?
3 R/S →	YES
	14 - 7 = ?
6 R/S →	NO 14 - 7 = ?
8 R/S →	NO 14 - 7 = ?
	13 - 2 = ?
11 R/S →	YES
	14 - 13 = ?
1 R/S →	YES
	14 - 10 = ?
4 R/S →	YES
	12 - 9 = ?
3 R/S →	YES
	14 - 10 = ?
4 R/S →	YES
	90% RIGHT
	+, -, *, / ?

Astuce de programmation

Ce programme illustre une combinaison de diverses caractéristiques alphanumériques du HP-41C. Appels indirects de sous-programmes et utilisation de labels alphabétiques de sortie. Vous choisissez l'un des quatre opérateurs arithmétiques proposés. Le programme stocke cet opérateur dans le registre 06, génère deux nombres, puis exécute le sous-programme dont le nom était stocké dans R06. L'opérateur est ensuite rappelé à l'affichage pour formuler le problème que vous avez à résoudre.

Autre astuce

Une autre ingéniosité dans ce programme: le générateur de nombres aléatoires.

Ce générateur:

$$r_{n+1} = \text{FRC}(9821 \times r_n + 0,211327)$$

découvert par un utilisateur du HP-65 (il passe avec succès le test spectral), permet d'obtenir un million de nombres aléatoires différents compris entre 0 et 1, quelle que soit la valeur choisie pour r_0 (racine).

Pour obtenir les nombres aléatoires entiers servant aux exercices, il faut effectuer sur ces nombres compris entre 0 et 1, une petite transformation: il suffit de les multiplier par un entier N , puis de prendre la partie entière du résultat pour obtenir des nombres aléatoires compris entre 0 et $N-1$. Ce programme utilise $N =$ votre valeur maximale désirée $+ 1$, pour que le générateur vous donne des nombres compris entre 0 et la valeur maximale choisie.

<pre> 01+LBL "TEA CH" 02 CF 29 03 FIX 0 04 STO 00 05+LBL A 06 "MAX NUM BER?" 07 PROMPT 08 1 09 + 10 STO 04 11+LBL "AGN " 12 0 13 STO 08 14 STO 09 15 10 16 STO 07 17 "+, -, * ./?" 18 AON 19 PROMPT 20 AOFF 21 ASTO 06 22+LBL 09 23 XEQ "RND M" 24 STO 02 25 XEQ "RND M" 26 STO 05 27 RCL 02 28 XEQ IND 06 29+LBL "TRY " 30 ARCL 05 31 ARCL 06 32 ARCL 02 33 "I=?" 34 PROMPT 35 RCL 03 36 X=Y? 37 GTO "YES " 38 "NO " 39 AVIEW 40 TONE 2 41 TONE 2 </pre>	<p>Initialisation.</p> <p>Demande du nombre maximum.</p> <p>Label de début d'une série d'exercices.</p> <p>Demande le type d'opération.</p> <p>Début de la boucle.</p> <p>Génération des opérandes.</p> <p>Génération du problème.</p> <p>Pose du problème.</p> <p>Si la réponse est correcte, affichage de OUI.</p>	<pre> 42 FS?C 00 43 GTO 00 44 SF 00 45 1 46 ST+ 09 47 GTO "TRY " 48+LBL 00 49 ARCL 05 50 ARCL 06 51 ARCL 02 52 "I=" 53 ARCL 03 54 AVIEW 55 GTO 00 56+LBL "YES " 57 CF 00 58 "YES" 59 AVIEW 60 1 61 ST+ 08 62+LBL 00 63 DSE 07 64 GTO 09 65 RCL 09 66 X=0? 67 XEQ "FF" 68 RCL 08 69 .1 70 / 71 CLA 72 ARCL X 73 "I% RIGH T" 74 AVIEW 75 PSE 76 PSE 77 GTO "AGN " 78+LBL "+ " 79 + 80 STO 03 81 LASTX 82 - 83 LASTX 84 CLA 85 RTN 86+LBL "- " 87 - </pre>	<p>S'il s'agit du 2^e essai, passage au problème suivant.</p> <p>Comptage de la mauvaise réponse et répétition du problème.</p> <p>Affichage de la réponse juste.</p> <p>Affichage de OUI. Comptage de la réponse juste.</p> <p>S'il reste des problèmes alors répétition de la boucle. S'il n'y a pas eu de mauvaise réponse, alors tonalité musicale.</p> <p>Affichage du score (en %).</p> <p>Reprise d'une nouvelle série d'exercices. Problèmes d'addition.</p> <p>Problèmes de soustraction.</p>
<p>R00=Nombre aléatoire R01=Inutilisé R02=n2 R03=Réponse R04=1+nombre maximum</p>		<p>R05=n1 R06=Type de problème R07=Compteur R08=Compteur de réponses justes R09=Compteur de mauvaises réponses</p>	

<pre> 88 X<=0? 89 XEQ 00 90 STO 03 91 LASTX 92 + 93 LASTX 94 CLA 95 RTN 96♦LBL 00 97 CHS 98 RCL 02 99 X<> 05 100 X<> 02 101 RDN 102 RTN 103♦LBL "*" 104 * 105 STO 03 106 LASTX 107 / 108 LASTX 109 CLA 110 RTN 111♦LBL " / " 112 X<>Y 113 STO 03 114 * 115 STO 05 116 CLA 117 RTN 118♦LBL "RND M" 119 RCL 00 120 9821 121 * 122 .211327 123 + 124 FRC 125 STO 00 126 SQRT 127 RCL 04 128 * 129 INT 130 RTN 131♦LBL "FF" 132 TONE 8 133 TONE 9 134 XEQ "0" 135 XEQ "0" 136 TONE 8 </pre>	<p>Problèmes de multiplication.</p> <p>Problèmes de division.</p> <p>Générateur de nombres aléatoires.</p> <p>Construit les nombres.</p> <p>Emission d'une tonalité musicale.</p>	<pre> 137 TONE 8 138 TONE 8 139 TONE 7 140 TONE 8 141 TONE 8 142 TONE 7 143 TONE 8 144 TONE 9 145 XEQ "0" 146 XEQ "0" 147 TONE 9 148 TONE 8 149 XEQ "0" 150 TONE 8 151 TONE 7 152 XEQ "0" 153 TONE 7 154 TONE 6 155 RTN 156♦LBL "0" 157 X<>Y 158 X<>Y 159 X<>Y 160 X<>Y 161 X<>Y 162 X<>Y 163 RTN </pre> <p>Etat Size: 010 Fix 0 CF 29</p> <p>F00 armé si reponse incorrecte. F29 desarmé pour annuler le point décimal.</p>	<p>Sous-programme à utiliser pour un minutage.</p>


Conversion hexadécimale/décimale

Ce programme effectue les conversions entre les nombres entiers décimaux inférieurs à 1048575 et les nombres entiers hexadécimaux inférieurs à FFFFF.

				Taille : 021
N°	Instructions	Données	Fonctions	Affichage
1	Introduisez ce programme, puis mettez-vous en mode personnel.			
2	Initialisez le programme.		[A]	READY
3	Pour convertir un nombre décimal en hexadécimal, introduisez le nombre.	D	[E]	H
4	Pour convertir un nombre hexadécimal en décimal, introduisez-le en mode ALPHA.	H	[E]	D
5	Pour reconverter le nombre, appuyez de nouveau sur E. Note : D représente un entier décimal inférieur à 1048575 (base 10), H représente un entier hexadécimal inférieur à 1000000 (base 16).			H ou D

Exemple 1 :

Calculez l'équivalent hexadécimal de 123 décimal.



Exemple 2 :

Calculez l'équivalent décimal de 123 hexadécimal.



Astuce de programmation

Ce programme utilise les indicateurs d'entrée numérique et d'entrée alphanumérique (indicateurs 22 et 23) qui déterminent si votre nombre est décimal ou hexadécimal. La première ligne du programme teste l'indicateur 22. Si l'entrée est numérique, l'indicateur 23 est désarmé et le programme va directement à l'opération 5. Si l'indicateur 22 n'est pas armé, l'indicateur 23 est testé, provoquant un branchement à LBL04 si les données introduites étaient alphanumériques. A la fin du programme ces indicateurs sont automatiquement ajustés pour permettre la conversion inverse.

<p>01+LBL E 02 FS?C 22 03 CF 23 04 FS? 23 05 GTO 04 06 STO 19 07 XEQ 08 08 + 09+LBL 01 10 LASTX 11 ISG 16 12+LBL 00 13 1 E2 14 / 15 INT 16 X#0? 17 GTO 01 18 CLA 19 LASTX 20+LBL 03 21 1 E2 22 * 23 ARCL IND X 24 FRC 25 DSE 16 26 GTO 03 27 SF 23 28 ASTO X 29 BEEP 30 RTN 31+LBL 04 32 ASTO 16 33 .00002 34 STO 17 35 0 36 STO 19 37+LBL 05 38 0 39 STO 18 40 " " 41 ASTO Y 42 ARCL 16 43 ASTO 16 44 ASHF 45 ASTO X 46 X=Y? 47 GTO 08 48 CLA 49 ASTO Y</p>	<p>Si donnée alpha-numérique, branchement au label 04. Conversion décimale en hexadécimale. Boucle 1.</p> <p>Comptage avec incrément. Label fictif.</p> <p>Tant qu'il reste des chiffres à traiter, passage de la boucle 1. Début de la boucle 3. Construction du nombre hex.</p> <p>Répétition de la boucle 3 jusqu'à ce que R16 soit vidé. Affichage du nombre hex.</p> <p>Préparation à la conversion hexadécimale/décimale.</p> <p>Début de la boucle 5.</p> <p>Séparation des éléments du nombre hexadécimal.</p> <p>Si le caractère est blanc, sortie de la boucle.</p>	<p>50 X=Y? 51 GTO 05 52+LBL 06 53 RCL IND 18 54 X=Y? 55 GTO 07 56 RDH 57 ISG 18 58+LBL 00 59 GTO 06 60+LBL 07 61 RCL 18 62 RCL 17 63 INT 64 10↑X 65 * 66 ST+ 19 67 ISG 17 68 GTO 05 69+LBL 08 70 16 71 STO 18 72 1 73 STO 17 74 0 75 STO 16 76 1 E2 77 STO 20 78 FS? 23 79 GTO 09 80 RCL 18 81 X<> 20 82 STO 18 83+LBL 09 84 RCL 19 85+LBL 10 86 RCL 20 87 / 88 STO 19 89 FRC 90 RCL 20 91 * 92 RCL 17 93 * 94 ST+ 16 95 RCL 18 96 ST* 17 97 RCL 19 98 INT 99 X#0?</p>	<p>Si le caractère est nul, répétition de la boucle 5.</p> <p>Construction du nombre codé hexadécimal.</p> <p>Comptage de 5 caractères hexadécimaux au max. Routine de stockage de constantes dans les registres appropriés et mise en place de la conversion.</p> <p>Début de la boucle 10. Conversion d'un nombre d'une base dans l'autre.</p> <p>Si non effectué,</p>
<p>R00=«0» R01=«1» R02=«2» R03=«3» R04=«4» R05=«5»</p>		<p>R06=«6» R07=«7» R08=«8» R09=«9» R10=«A» R11=«B»</p>	


<pre> 100 GTO 10 101 X<> 16 102 CLA 103 FS?C 23 104 BEEP 105 RTN 106+LBL A 107 CF 22 108 CF 23 109 "0" 110 ASTO 00 111 "1" 112 ASTO 01 113 "2" 114 ASTO 02 115 "3" 116 ASTO 03 117 "4" 118 ASTO 04 119 "5" 120 ASTO 05 121 "6" 122 ASTO 06 123 "7" 124 ASTO 07 125 "8" 126 ASTO 08 127 "9" 128 ASTO 09 129 "A" 130 ASTO 10 131 "B" 132 ASTO 11 133 "C" 134 ASTO 12 135 "D" 136 ASTO 13 137 "E" 138 ASTO 14 139 "F" 140 ASTO 15 141 "READY" 142 ASTO X </pre>	<p>alors répétition de la boucle 10.</p> <p>Routine d'initialisation.</p>		
<p>Etat: Size: 021 Fix 0</p> <p>Indic F22 Entrées numériques F23 Entées alphanumériques</p> <p>R12=«C» R13=«D» R14=«E» R15=«F» R16=Alphanumérique</p> <p>R17=Compteur de boucle, compteur de chiffres R18=Base (constante), compteur de boucle R19=Construction du nombre codé décimal R20=Base (constante)</p>			

Calculs financiers

Ce programme permet de résoudre de nombreux problèmes dans lesquels interviennent les paramètres argent, temps et intérêt.

Tout problème d'annuités ou de capitalisation, peut être représenté par un diagramme du type suivant:

Argent reçu:
positif
(flèche vers
le haut).



Argent versé:
négatif
(flèche vers
le bas).

où:

PV (present value) = valeur actuelle: emprunt, prêt, investissement.

I = taux d'intérêt périodique.

N = nombre de périodes.

PMT (payment) = remboursement d'un emprunt, remboursement d'un prêt, rendement d'un investissement.


FV (future value) = valeur: valeur future, valeur résiduelle

Le diagramme ci-dessus décrit par exemple l'amortissement d'un emprunt. Du point de vue du prêteur, le schéma est inversé: PV est négative et les PMT sont positifs.

On peut simuler n'importe quel type de transaction sur ce type de diagramme, en changeant simplement les signes de PV, PMT et FV. Ci-dessous figurent les diagrammes des quatre problèmes élémentaires d'intérêts composés.

Le programme ayant été développé aux Etats-Unis, il affiche les différentes sommes précédées du signe \$. Celui-ci peut être comme F pour francs, les calculs n'en sont pas modifiés.

Diagrammes généraux de flux financiers et terminologie.



Les cinq touches de la rangée supérieure (A et E) correspondent respectivement à N , I , PV , PMT et FV . Elles servent à introduire et à calculer ces paramètres financiers. Si vous introduisez trois de ces cinq paramètres, il vous suffit d'appuyer sur l'une des deux touches restantes pour connaître la valeur correspondante. Pour résoudre un nouveau problème, il n'est pas nécessaire de réintroduire les valeurs qui n'ont pas changé. Pour afficher chacune des différentes valeurs, il suffit d'appuyer sur **RCL** puis sur la touche désirée. La séquence **■** **A** permet d'effacer tous les registres mobilisés par ce programme. Dans ce cas, le message N , I , PV , PMT , FV apparaît sur l'écran pour vous rappeler les fonctions des cinq touches.

				Taille : 010
N°	Instructions	Données	Fonctions	Affichage
1	Introduisez le programme, et placez ensuite le calculateur en mode personnel.			
2	Effacez les registres financiers.		■ A	N, I, PV, PMT, FV
3	Enregistrez vos données			
	nombre de périodes	N	A	N
	taux d'intérêt périodique (en pour-cent)	I	B	I
	valeur actuelle de l'investissement	PV	C	PV
	paiement périodique	PMT	D	PMT
	valeur capitalisée de l'investissement	FV	E	FV
4	Calculez votre inconnue			
	nombre de périodes		A	N=N
	taux d'intérêt périodique		B	I=I%
	valeur actuelle de l'investissement		C	PV=\$PV
	paiement périodique		D	PMT=\$PMT
	valeur capitalisée de l'investissement		E	FV=\$FV
5	Vous pouvez retourner à l'instruction n° 4 pour recalculer l'une quelconque des cinq valeurs, ou à l'instruction n° 3 pour modifier l'une ou plusieurs d'entre elles.			

Exemple 1 :

Un achat de 50 000 \$ couvert par un emprunt de 40 000 \$ à 8,5% d'intérêts, et remboursable sur 30 ans moins un mois. Quel est le montant des mensualités?

Appuyez sur	Affichage
■ A 40000 C →	40.000,00
8.5 ENTER 12 + B →	0.71
30 ENTER 12 x 1 - A D →	PTM=\$-307,75

Exemple 2 :

La vente de cet achat 18 mois plus tard, produit un résultat net de 25 000 \$. A quel taux d'intérêt aurait-il fallu placer le capital initial de 10 000 \$ et l'épargne mensuelle de 307,75 pour obtenir un capital de 25 000 \$ au bout de 18 mois?

Appuyez sur	Affichage	
18 A		
25000 E →	25.000,00	
10000 CHS C B →	I=3.21%	taux mensuel
12 x →	38.51	taux annuel

Astuce de programmation

Ce programme illustre une technique particulière qu'on appelle «solution interchangeable». Chacune des cinq variables de l'équation peut être formulée en fonction des quatres autres. Chacune des cinq touches de la rangée supérieure est utilisée aussi bien pour introduire les données que pour calculer les inconnues, grâce à l'organisation suivante du programme :

LBLT	L'une des touches A à J (ou a à e).
STOr	Stockage de la variable dans le registre r.
FS?C22	Teste l'indicateur d'entrée numérique et le désarme.
RTN	Arrêt si cette donnée est nouvelle.
}	Calcul de l'inconnue.
STOr	Stockage de la valeur calculée.
}	Affichage de la nouvelle valeur.
RTN	

Ce bloc d'instructions peut être répété pour chacune des variables.

01♦LBL A		51 ABS	
02 STO 01	Stockage de N.	52 RCL 04	
03 FS?C 22	Si nouvelle donnée,	53 RCL 01	
04 RTN	alors arrêt,	54 *	
05 RCL 04	sinon calcul du	55 RCL 03	
06 RCL 09	nouvel N.	56 +	
07 /		57 ABS	
08 STO 00		58 -	
09 RCL 05		59 RCL 04	
10 -		60 RCL 01	
11 RCL 03		61 *	
12 RCL 00		62 RCL 05	
13 +		63 +	
14 /		64 ABS	
15 LN		65 RCL 03	
16 RCL 09		66 ABS	
17 LN1+X		67 -	
18 /		68 *	
19 STO 01		69 ENTER↑	
20 "N="	Affichage du nouvel	70 ABS	
21 ARCL X	N.	71 /	
22 AVIEW		72 1 E-9	
23 RTN		73 *	
24♦LBL B		74 STO 09	
25 STO 02	Stockage de I et de	75♦LBL 06	Début de la boucle.
26 1 E2	certaines fonctions	76 XEQ 08	
27 /	de I.	77 RCL 04	
28 STO 09		78 *	
29 1		79 RCL 03	
30 +		80 +	
31 STO 07		81 RCL 05	
32 RCL 02		82 RCL 08	
33 FS?C 22	Si nouvelle donnée,	83 *	
34 RTN	alors arrêt	84 +	
35 RCL 04	sinon,	85 RCL 08	
36 X≠0?	si PMT=0	86 RCL 07	
37 GTO 01	calcul de I par	87 /	
38 RCL 05	la formule simple.	88 RCL 01	
39 RCL 03		89 *	
40 /		90 STO 06	
41 CHS		91 1	
42 RCL 01		92 RCL 08	
43 1/X		93 -	
44 Y↑X		94 RCL 09	
45 1		95 /	
46 -		96 -	
47 STO 09		97 RCL 04	
48 GTO 00		98 RCL 09	
49♦LBL 01		99 /	
50 RCL 05	Si non calcul de I	100 *	
	par la formule de	101 RCL 05	
	Newton.		
R00=Utilisé		R06=Utilisé	
R01=n		R07=1+i/100	
R02=i		R08=Utilisé	
R03=PV		R09=i/100	
R04=PMT			
R05=FV			

<pre> 102 RCL 05 103 * 104 - 105 / 106 ST- 09 107 ABS 108 1 E-7 109 X<=Y? 110 STO 06 111 RCL 09 112 LBL 00 113 1 E2 114 * 115 STO 02 116 "I=" 117 ARCL X 118 "I%" 119 AVIEW 120 RTN 121 LBL 0 122 STO 03 123 FS?C 22 124 RTN 125 RCL 04 126 XEQ 08 127 * 128 RCL 05 129 RCL 08 130 * 131 + 132 CHS 133 STO 03 134 "PV=#" 135 ARCL X 136 AVIEW 137 RTN 138 LBL 0 139 STO 04 140 FS?C 22 141 RTN 142 XEQ 08 143 1/X 144 RCL 03 145 RCL 05 146 RCL 08 147 * 148 + 149 * 150 CHS 151 STO 04 152 "PMT=#" 153 ARCL X </pre>	<p>Si I n'est pas plus petit, répétition de la boucle.</p> <p>Affichage du nouvel I.</p> <p>Stockage de PV. Si nouvelle donnée, alors arrêt, sinon calcul de la nouvelle PV.</p> <p>Affichage de la nouvelle PV.</p> <p>Stockage de PMT. Si nouvelle valeur, alors arrêt, sinon calcul du nouveau PMT.</p> <p>Affichage du nouveau PMT.</p>	<pre> 154 AVIEW 155 RTN 156 LBL E 157 STO 05 158 FS?C 22 159 RTN 160 XEQ 08 161 RCL 04 162 * 163 RCL 03 164 + 165 RCL 08 166 / 167 CHS 168 STO 05 169 "FV=#" 170 ARCL X 171 AVIEW 172 RTN 173 LBL 00 174 1 175 XEQ 09 176 RCL 01 177 CHS 178 Y↑X 179 STO 08 180 - 181 RCL 09 182 / 183 RTN 184 LBL 09 185 RCL 09 186 1 187 + 188 STO 07 189 RTN 190 LBL a 191 CLX 192 STO 01 193 STO 02 194 STO 03 195 STO 04 196 STO 05 197 STO 09 198 "N, I, P v, PMT, F" 199 "FV" 200 AVIEW 201 RTN </pre>	<p>Stockage de FV. Si nouvelle donnée, alors arrêt, sinon calcul de la nouvelle FV.</p> <p>Sous-programme pour le calcul</p> <p>de: $\left(1 + \frac{i}{100}\right)^{-n}$</p> <p>et: $\left(1 + \frac{i}{100}\right)^n - 1$</p> <p>Sous-programme pour le calcul de $1 + 1/100$.</p>
--	---	---	---

Etat:
Size=010
Fix 02
F22 Entrées
numériques

Recherche de racines

La recherche de racines est le calcul de valeurs, qui, affectées à une variable indépendante, rendent nulle une fonction particulière de cette variable.

Par exemple: $x = 3$ est une racine pour l'équation $f(x) = 2x - 6$ parce que $f(3) = (2 \times 3) - 6 = 0$

Il existe plusieurs méthodes de recherche des racines de $f(x)$. L'algorithme, par exemple, commence par une première valeur d'essai et procède par itération jusqu'à obtention d'une solution. Certains algorithmes n'aboutissent pas à une solution et le processus d'itération est infini. D'autres, par contre, sont très longs, mais aboutissent inévitablement à une solution. L'algorithme utilisé dans ce programme donnera toujours une racine, à partir de valeurs d'essai initiales encadrant un nombre impair de racines. Dans le cas contraire, il faudra choisir de nouvelles valeurs d'essai. Le tribut que vous devez payer en échange de la certitude d'un résultat est une bonne connaissance de votre fonction avant l'utilisation de ce programme.

Avant d'exécuter ce programme de recherche des racines, il faut programmer la fonction dont vous désirez calculer les racines: appuyez sur **GTO** \square \square puis chargez votre programme. Exécutez ensuite **XEQ** **RACINE** pour commencer le programme. Celui-ci vous demande d'introduire le nom que vous attribuez à votre fonction (NOM FONCTION), puis les deux valeurs d'essai initiales (ESSAI 1, ESSAI 2). Si pour ces deux valeurs, la fonction a le même signe, le message « $F1 * F2 > 0$ » apparaît sur l'écran, aussitôt remplacé par une nouvelle demande de valeurs d'essai.


Ce programme mobilise les registre 01 à 07, si bien que tous les autres registres (R00, R08 et suivants) sont disponibles pour l'évaluation de votre fonction. Une réponse étiquetée est affichée lorsque la valeur de la fonction est inférieure à 10^{-10} . Pour obtenir une précision supérieure vous pouvez recommencer le programme en introduisant une valeur légèrement différente.

				Taille: 008
N°	Instructions	Données	Fonctions	Affichage
1	Introduisez le programme.			
2	Introduisez votre fonction en lui donnant un nom autorisé (ni A à J, ni a à e, ni 00 à 99).			
3	Commencez l'exécution du programme.		XEQ RACINE	FUNCTION NAME?
4	Introduisez le nom de la fonction.	NOM	R/S	GUESS 1=?
5	Introduisez la 1 ^{re} valeur d'essai.	X1	R/S	GUESS 2=?
6	Introduisez la 2 ^e valeur d'essai: – ou bien une racine affichée – ou bien le programme retourne à l'instruction n° 5.	X2	R/S	X=(racine) F1* F2 > 0

Exemple 1:

Pour quelle valeur de x, l'équation $R(x) = x^3 - 6x^2 + 11x - 1$ est-elle égale à 0? Un examen rapide de la fonction indique une racine comprise entre 0 et 1.

Appuyez sur	Affichage
GTO PRGM	
LBL ALPHA R ALPHA	
ENTER ENTER ENTER 6 - X	
11 + X 1 - RTN	
PRGM	
XEQ ALPHA ROOT ALPHA	FUNCTION NAME?
R R/S	GUESS1=?
0 R/S	GUESS2=?
1 R/S	X=0.0958




Exemple 2:

Trouvez une racine pour

$$W(x) = \sin(x - 30) - \cos(2x + 60)$$

qui soit comprise entre 200 et 300 degrés.

Appuyez sur		Affichage	
	GTO		PRGM
	LBL		ALPHA WAVE
	SIN		RCL 04
2		60	
			RTN
	PRGM		
	XEQ		ALPHA ROOT
			ALPHA
			FUNCTION NAME?
			GUESS1 = ?
200			R/S
			GUESS2 = ?
300			R/S
			X=260.0000



Astuce de programmation

Le programme de recherche de racines vous demande d'introduire le nom de votre fonction. Il enregistre ce nom, puis exécute indirectement la fonction. Vous remarquerez que la fonction AON est exécutée avant PROMPT, si bien que le HP-41C s'arrête en mode ALPHA. La fonction AOFF doit cependant être exécutée avant le PROMPT suivant, sinon le mode ALPHA reste opératoire. AON et AOFF permettent de contrôler le mode dans lequel s'arrête le calculateur, afin de vous rappeler par la suite quel type de données vous devez envoyer.

Le nom de votre fonction figurant dans le registre 3, le programme peut exécuter celle-ci à tout moment, grâce à : XEQ IND 03. Par conséquent, un programme qui aurait nécessité normalement des modifications pour chacune des fonctions utilisées, ne demande, grâce à cette astuce, plus que le nom des différentes fonctions.

- | | |
|----------------|--|
| FUNCTION NAME? | |
| AON | |
| PROMPT | Affichage du message avec arrêt en mode ALPHA. |
| ASTO 03 | Stockage du nom dans R3. |
| AOFF | Suppression du mode ALPHA. |
| XEQ IND 03 | Exécution du programme dont le nom est stocké dans R3. |

<pre> 01♦LBL "ROOT" T" 02 "FUNCTION NAME?" 03 AON 04 PROMPT 05 AOFF 06 ASTO 03 07♦LBL A 08 "GUESS1= ?" 09 PROMPT 10 STO 01 11 "GUESS2= ?" 12 PROMPT 13 STO 02 14 RCL 01 15 STO 04 16 XEQ IND 03 17 STO 05 18 RCL 02 19 STO 04 20 XEQ IND 03 21 STO 06 22 RCL 05 23 * 24 X>0? 25 GTO 05 26♦LBL 00 27 RCL 02 28 RCL 02 29 RCL 01 30 - 31 RCL 06 32 RCL 05 33 - 34 / 35 RCL 06 36 * 37 - 38 STO 04 39 XEQ IND 03 40 STO 07 41 X=0? 42 GTO 04 43 ABS </pre>	<p>Demande du nom de la fonction.</p> <p>Stockage des valeurs d'essai.</p> <p>Début de la boucle.</p> <p>Nouvel x.</p> <p>Si $f(x)=0$ la racine est trouvée.</p>	<pre> 44 1 E-10 45 X>Y? 46 GTO 04 47 RCL 07 48 RCL 06 49 * 50 X>0? 51 GTO 01 52 RCL 02 53 STO 01 54 RCL 06 55 STO 05 56♦LBL 02 57 RCL 04 58 STO 02 59 RCL 07 60 STO 06 61 GTO 00 62♦LBL 01 63 2 64 ST/ 05 65 GTO 02 66♦LBL 04 67 "X=" 68 ARCL 04 69 PROMPT 70♦LBL 05 71 "F1*F2>0 " 72 AVIEW 73 PSE 74 GTO A </pre> <p>Etat: Size=008 DEG Fix 04</p>	<p>Valeur de tolérance si $f(x) < 1E-10$ la racine est trouvée.</p> <p>Choix de nouvelles valeurs d'essai.</p> <p>Racine trouvée.</p> <p>Affichage de la réponse.</p> <p>Message d'erreur.</p> <p>Retour à l'introduction.</p>
<p>R00=Inutilisé R01=x1 R02=x2 R03=Nom R04=x R05=f(x1) R06=f(x2) R07=f(x3)</p>			

Ajustement de courbes

Ce programme permet d'ajuster les données (x_i, y_i) , d'un ensemble de i observations statistiques ($i=1, 2, \dots, n$), à l'une quelconque des courbes suivantes :

1. Droite (régression linéaire): $y = a + bx$
2. Exponentielle: $y = ae^{bx}$ ($a > 0$)
3. Logarithmique: $y = a + b \ln x$
4. Fonction puissance: $y = ax^b$ ($a > 0$).

Les coefficients de régression a et b sont les solutions du système d'équations linéaires suivants:

$$An + B\sum X_i = \sum Y_i$$

$$A\sum X_i + B\sum X_i^2 = \sum Y_i X_i$$


Lorsque les relations entre les variables sont définies comme ci-dessous :

Regression	A	B	X_1	Y_1
Lineaire	a	b	x_1	y_1
Exponentielle	$\ln a$	b	x_1	$\ln y_1$
Logarithmique	a	b	$\ln x_1$	y_1
Puissance	$\ln a$	b	$\ln x_1$	$\ln y_1$


Le coefficient de détermination est:

$$R^2 = \frac{A\sum Y_i + b\sum X_i Y_i - \frac{1}{n} (\sum Y_i)^2}{\sum (Y_i^2) - \frac{1}{n} (\sum Y_i)^2}$$

Régression linéaire




Ajustement par une courbe exponentielle




46 Ajustement de courbes

Ajustement par une courbe logarithmique



Ajustement par une courbe de fonction puissance



Remarques:

1. Ce programme applique la méthode des moindres carrés, soit aux équations de départ (droite ou courbe logarithmique) soit aux équations transformées (courbe exponentielle et courbe de fonction puissance).
2. Une valeur négative ou nulle pour x_i provoque une erreur dans le cas d'un ajustement à une courbe logarithmique. Une valeur négative ou nulle pour y_i provoque une erreur dans le cas d'un ajustement à une courbe exponentielle. Pour un ajustement à une courbe de fonction puissance, x_i et y_i doivent être strictement positifs.
3. Plus les écarts entre les x_i (et/ou les y_i) sont petits, plus la précision des coefficients de régression diminue.

				Taille: 016
N°	Instructions	Données	Fonctions	Affichage
1	Introduisez le programme.			
2	Initialisez le programme pour une droite pour une exponentielle pour une courbe logarithme pour une courbe puissance		XEQ LIN XEQ EXP XEQ LOG XEQ POW	LIN EXP LOG POW
3	Répétez les n° 3 et 4 pour $i=1, 2, \dots$ n valeurs: x_i y_i	x_i y_i	ENTER A	(i)
4	Correction en cas d'erreur d'introduction.	x_k y_k	ENTER C	(k-1)
5	Calcul de R^2 et des coefficients de régression.		E R/S R/S	$R^2 = (R^2)$ $a = (a)$ $b = (b)$
6	Calcul de y estimé pour x.	x	R/S	$y = (\hat{y})$
7	Répétez le n° 5 pour différents x.			
8	Répétez le n° 4 pour réafficher les résultats.			
9	Pour un nouveau jeu de données puis allez en 3.		 A	LIN ou EXP ou LOG ou POW
10	Pour utiliser un autre programme allez en 2.			

Exemple 1:

Ajustez le jeu de données suivant à une droite, puis calculez \hat{y} pour $x = 37$ et $x = 35$.

x_i	40,5	38,6	37,9	36,2	35,1	34,6
y_i	104,5	102	100	97,5	95,5	94

Appuyez sur**Affichage**

XEQ ALPHA	LIN	ALPHA	→	LIN
40.5	ENTER	104.5	A	1.00
38.6	ENTER	102	A	2.00
37.9	ENTER	100	A	3.00
36.2	ENTER	97.5	A	4.00
35.2	ENTER	95.5	A	5.00
35.2	ENTER	95.5	C	4.00
35.1	ENTER	95.5	A	5.00
34.6	ENTER	94	A	6.00
E				R2=0.99
R/S				a=33.53
R/S				b=1.76
37	R/S			Y.=98.65
35	R/S			Y.=95.13

Corrigez l'erreur.
Utilisez les valeurs correctes

Solution:

$a = 33,53, b = 1,76$

$R^2 = 0,99$

L'équation de la droite est: $y = 33,53 + 1,76x$

Lorsque $x = 37, \hat{y} = 98,65$

Lorsque $x = 35, \hat{y} = 95,13$

Exemple 2:

Ajustez le jeu de données suivant à une courbe exponentielle, puis calculez \hat{y} pour $x = 1,5$ et $x = 2$.

x_i	0,72	1,31	1,95	2,58	3,14
y_i	2,16	1,61	1,16	0,85	0,5

Appuyez sur	Affichage
XEQ ALPHA SIZE ALPHA 016	
XEQ ALPHA EXP ALPHA →	EXP
.72 ENTER ↑ 2.16 A →	1.00
1.31 ENTER ↑ 1.61 A →	2.00
1.95 ENTER ↑ 1.16 A →	3.00
2.58 ENTER ↑ .85 A →	4.00
3.15 ENTER ↑ .05 A →	5.00
3.15 ENTER ↑ .05 C →	4.00
3.14 ENTER ↑ 0.5 A →	5.00
E →	R2=0.98
R/S →	a=3.45
R/S →	b=-0.58
1.5 R/S →	Y.=1.44
2.0 R/S →	Y.=1.08

Si vous ne faites pas d'erreurs, vous pouvez sauter deux étapes.

Solution:

$a = 3,45, b = -0,58$

L'équation de la courbe est: $y = 3,45e^{-0,58x}$

$R^2 = 0,98$

Lorsque $x = 1,5, y = 1,44$

Lorsque $x = 2, \hat{y} = 1,08$

Exemple 3:

Ajustez le jeu de données suivant à une courbe logarithmique et calculez \hat{y} pour $x = 8$ et $x = 14,5$.

x_i	3	4	6	10	12
y_i	1,5	9,3	23,4	45,8	60,1

Appuyez sur	Affichage
XEQ ALPHA SIZE ALPHA 016	
XEQ ALPHA LOG ALPHA	LOG
3 ENTER 1.5 A	1.00
4 ENTER 9.3 A	2.00
6 ENTER 23.4 A	3.00
10 ENTER 45.8 A	4.00
12 ENTER 6.01 A	5.00
12 ENTER 6.01 C	4.00
12 ENTER 60.1 A	5.00
E	R ² =0.98
R/S	a=-47.02
R/S	b=41.39
8 R/S	Y.=39.06
14.5 R/S	Y.=63.67

Une autre erreur!

Solution:

$$a = -47,02, b = 41,39$$

L'équation de la courbe est : $y = 47,02 + 41,39 \ln x$

$$R^2 = 0,98$$

Lorsque $x = 8$, $\hat{y} = 39,06$

Lorsque $x = 14,5$, $\hat{y} = 63,67$

Exemple 4:

Ajustez le jeu de données suivant à une courbe de fonction puissance et calculez \hat{y} pour $x = 18$ et $x = 23$.

x_i	10	12	15	17	20	22	25	27	30	32	35
y_i	0,95	1,05	1,25	1,41	1,73	2,00	2,53	2,98	3,85	4,59	6,02

Appuyez sur	Affichage
XEQ ALPHA SIZE ALPHA 016	
XEQ ALPHA POW ALPHA	POW
10 ENTER 0.95 A	1.00
12 ENTER 1.05 A	2.00
15 ENTER 1.25 A	3.00
17 ENTER 1.41 A	4.00
20 ENTER 1.73 A	5.00
22 ENTER 2.00 A	6.00
25 ENTER 2.53 A	7.00
27 ENTER 2.98 A	8.00
30 ENTER 3.85 A	9.00
32 ENTER 4.59 A	10.00
35 ENTER 60.2 A	11.00
35 ENTER 60.2 C	10.00
35 ENTER 6.02 A	11.00
E	R ² =0.94
R/S	a=0.03
R/S	b=1.46
18 R/S	Y.=1.76
23 R/S	Y.=2.52

Nouvelle correction d'erreurs.

Solution:

$$a = 0,03, b = 1,46$$

L'équation de la courbe est: $y = 0,03x^{1,46}$

$$R^2 = 0,94$$

Lorsque $x = 18$, $\hat{y} = 1,76$

Lorsque $x = 23$, $\hat{y} = 2,52$

Astuce de programmation

Ce programme utilise un seul code pour la plupart des calculs qu'il doit faire. Chacun des quatre types de courbes nécessite une formulation différente des données: cela conduit normalement à utiliser quatre programmes d'ajustement différents. L'originalité, dans ce programme, a été d'adopter un traitement unique pour les quatre types de courbes: chacune de ces fonctions (LIN, LOG, EXP et POW) stocke un code dans le registre 00 et l'instruction XEQ IND 00 (ligne 32) gère les différents types de traitement des données, en exécutant la fonction dont le label est stocké dans le registre 00.

<pre> 01+LBL "LIN " 02 5 03 "LIN" 04 GTO 13 05+LBL "EXP " 06 6 07 "EXP" 08 GTO 13 09+LBL "LOG " 10 7 11 "LOG" 12 GTO 13 13+LBL "POW " 14 8 15 "POW" 16+LBL 13 17 XEQ "INI T" 18 STO 00 19 ASTO 08 20 ΣREG 10 21 CLΣ 22 BEEP 23 AVIEW 24 STOP 25+LBL C 26 X<>Y 27 XEQ IND 00 28 Σ- 29 STOP 30+LBL A 31 X<>Y 32 XEQ IND 00 33 Σ+ 34 STOP 35+LBL 07 36 LN 37 RTN 38+LBL 08 39 LN 40+LBL 06 41 X<>Y 42 LN 43 X<>Y 44 RTN </pre>	<p>Linéaire.</p> <p>Exponentielle.</p> <p>Logarithmique.</p> <p>Puissance.</p> <p>Signal sonore, affichage et initialisation des registres statistiques.</p> <p>Correction.</p> <p>Introduction de données.</p> <p>Log.</p> <p>Puissance et exp.</p>	<pre> 45+LBL E 46 RCL 15 47 RCL 11 48 RCL 10 49 RCL 10 50 XEQ 09 51 STO 03 52 RCL 12 53 RCL 11 54 RCL 10 55 RCL 14 56 XEQ 09 57 RCL 03 58 / 59 STO 04 60 XEQ IND 00 61 STO 06 62 RCL 15 63 RCL 14 64 RCL 10 65 RCL 12 66 XEQ 09 67 RCL 03 68 / 69 STO 05 70+LBL 03 71 RCL 04 72 RCL 12 73 * 74 RCL 05 75 RCL 14 76 * 77 + 78 RCL 12 79 x↑2 80 RCL 15 81 / 82 STO 09 83 - 84 RCL 13 85 RCL 09 86 - 87 / 88 "R2" 89 XEQ 88 90 RCL 06 91 "a" 92 XEQ 88 93 RCL 05 94 "b" </pre>	<p>Calcul de A, b et de a, b</p>
<pre> R00=Index R01=x R02=y R03=Coeff. de détermination R04=A </pre>	<pre> R05=b R06=a R07=Utilisé R08=LIN ou EXP ou LOG ou POW R09=(Σy) 2/n </pre>		

<pre> 95 GTO 01 96♦LBL 06 97♦LBL 08 98 E↑X 99♦LBL 05 100♦LBL 07 101 RTN 102♦LBL 09 103 * 104 STO 07 105 RDN 106 * 107 RCL 07 108 - 109 RTN 110♦LBL 00 111 "Y." 112♦LBL 01 113 "I=" 114 ARCL X 115 AVIEW 116 FS? 55 117 STOP 118♦LBL 04 119 GTO IND 00 120♦LBL 08 121 RCL 05 122 Y↑X 123 GTO 09 124♦LBL 06 125 RCL 05 126 * 127 E↑X 128♦LBL 09 129 RCL 06 130 * 131 GTO 00 132♦LBL 07 133 LN 134♦LBL 05 135 RCL 05 136 * 137 RCL 06 138 + 139 GTO 00 140♦LBL 08 141 "I=" 142 ARCL X 143 AVIEW 144 RTN </pre>	<p>Transformation inverse.</p> <p>Coeff. de détermination.</p> <p>Calcul de r^2.</p> <p>Introduction de x pour le calcul de y.</p>	<pre> 145♦LBL a 146 GTO IND 08 147♦LBL "INI T" 148 CLRG 149 CF 00 150 CF 01 151 CF 02 152 SF 21 153 SF 27 154 CF 29 155 RTN 156 END Etat: Size=16 Σ=10 Fix 02 F44 F41 F42 F21 F27 F29 </pre>	<p>Réinitialisation.</p> <p>Pour initialisation.</p>
<pre> R10=Σx R11=Σx² R12=Σy R13=Σy² R14=Σxy R15=n </pre>			

Calcul vectoriel

Ce programme effectue les opérations de base sur deux vecteurs, à savoir : addition, soustraction, multiplication et division.

Il faut placer les vecteurs dans la pile avant d'exécuter l'un des programmes.

Structure initiale de la pile

T u_1

Z u_1

Y u_2

X u_2

Affichage correspondant

U = u V = v

où les deux vecteurs sont :

$u_1 + iv_1$ et $u_2 + iv_2$.

On peut aussi utiliser la formulation suivante :

$u + vi$, $u + jv$ ou $ui + vj$.

				Taille: 000
N°	Instructions	Données	Fonctions	Affichage
1	Introduisez le programme et affectez les programmes que vous désirez CADD \oplus CSUB \ominus CMULT \otimes CDIV \oslash		ASN CADD \oplus ASN CSUB \ominus ASN CMULT \otimes ASN CDIV \oslash	
2	Placez vos données dans la pile opérationnelle Partie imaginaire du vecteur 1 Partie réelle du vecteur 1	V_1 U_1	ENTER \uparrow ENTER \uparrow	
2b	Partie imaginaire du vecteur 2 Partie réelle du vecteur 2	V_2 U_2	ENTER \uparrow	
3	Choisissez la fonction désirée Somme vectorielle Soustraction vectorielle Produit vectoriel Rapport vectoriel		CADD CSUB CMULT CDIV	$u = (u) \quad v = (v)$ $u = (u) \quad v = (v)$ $u = (u) \quad v = (v)$ $u = (u) \quad v = (v)$
4	Si vous voulez utiliser ce résultat pour un autre calcul, il n'est pas nécessaire de ré-introduire les données. Continuez simplement par l'opération n° 2b pour les vecteurs suivants.			

Exemple 1:Ajoutez $1 + i3$ à $4 + i6$.**Appuyez sur**■ **FIX** 2**Affichage**

Choisir une notation commode pour l'affichage des vecteurs.

6 **ENTER** 4 **ENTER** 3 **ENTER** 1
XEQ **ALPHA** CADD **ALPHA** → **U=5.00, V=9.00**

Exemple 2:Calculez $S^2 + 1$ pour $S = 3 + j2$.**Appuyez sur**2 **ENTER** 3 **ENTER**2 **ENTER** 3 **XEQ****ALPHA** CMULT **ALPHA** →**Affichage****U=5.00, V=12.00** 1+j00 **ENTER** 1 **XEQ****ALPHA** CADD **ALPHA** →**U=6.00, V=12.00****Astuce de programmation**

Les problèmes ne demandent en général qu'un seul nombre à l'utilisateur. Celui-ci introduit sa donnée, puis exécute sa fonction. Mais les vecteurs sont définis par deux nombres et il faut introduire deux vecteurs. Ce programme illustre parfaitement la réduction de longueur d'un programme lorsque l'utilisateur «économise» les données.

Vous remarquerez que si vous remplacez la partie du programme consacrée à la sortie par LBL «UV» RTN, les quatre routines peuvent être utilisées comme sous-routines dans tout programme faisant appel au calcul vectoriel. Les résultats u et v sont donnés respectivement dans les registres X et Y. Autre suggestion: il est commode d'affecter chacune des routines respectivement aux touches \oplus , \ominus , \otimes , et \oplus , afin d'exécuter immédiatement chaque fonction en mode personnel.

<pre> 01+LBL "CSU B" 02 CHS 03 X<>Y 04 CHS 05+LBL "CAD D" 06 X<>Y 07 RDN 08 + 09 RDN 10 + 11 R↑ 12 GTO "UV" 13+LBL "CDI V" 14 R-P 15 1/X 16 X<>Y 17 CHS 18 GTO 00 19+LBL "CMU LT" 20 R-P 21 X<>Y 22+LBL 00 23 RDN 24 RDN 25 R-P 26 R↑ 27 * 28 RDN 29 + 30 R↑ 31 P-R 32+LBL "UV" 33 "U=" 34 ARCL X 35 "F.V=" 36 ARCL Y 37 RVIEW 38 RTN Etat: Size=000 </pre>	<p>Soustraction.</p> <p>Changement de signe du second vecteur et addition.</p> <p>Division.</p> <p>Inversion du second vecteur et multiplication.</p> <p>Multiplication.</p> <p>Routine d'affichage.</p>		

Blackjack

Ce programme vous permet de jouer contre le calculateur à un jeu de cartes appelé «blackjack» (ou «21»). Le calculateur distribue (DL) des cartes à partir d'un sabot de deux jeux de 52 cartes (distribution exhaustive), et mélange des cartes (SH) lorsque toutes, sauf les 13 dernières, ont été distribuées. Le joueur peut faire n'importe quelle mise; s'il ne mise pas, le calculateur utilise sa mise précédente.

Règles du jeu

Le joueur et le meneur reçoivent chacun deux cartes. Le calculateur «retourne», pour exposer une de ses cartes. Le joueur peut alors tirer des cartes supplémentaires (HIT) ou passer (STAND). Le but du jeu est d'atteindre, sans jamais le dépasser, le score de 21 points, les figures comptant pour 10 points, l'as pour 1 ou 11 points et les autres cartes comptant pour leur valeur (2 à 10). Si les deux premières cartes du joueur totalisent 21, il fait «blackjack» et ramasse une fois et demie sa mise, à moins que le calculateur n'ait lui aussi fait «blackjack».

Un joueur qui tire une carte portant son total de points au-delà de 21, perd sa mise (BUST). Lorsque vous passez sur un total inférieur ou égal à 21, le calculateur doit tirer (HIT) jusqu'à ce que son total excède 16. A ce moment-là, c'est la main la plus forte qui gagne et votre banque (BANK) est remise à jour. Si vous avez le même total que le calculateur, la mise est remise en jeu.

Certaines des options du blackjack de casino ne sont pas couvertes par ce programme.

Il faut, pour exécuter ce programme, un module mémoire supplémentaire.

				Taille: 028
N°	Instructions	Données	Fonctions	Affichage
1	Chargez le programme et affectez DL (DEAL), HT (HIT) et S (STAND) aux touches que vous désirez. On peut placer une racine ($0 \leq \text{racine} < 1$) dans R00.			
2	Stockez votre banque initiale.	banque	[STO] 21	
3	Battez le sabot.		[XEQ] SH	SHUFFLING
4	Faites votre mise.	mise	DL	I SHOW C*
5a	Vous tirez des cartes (HIT), alors répétez cette même opération, ou allez à l'instruction 5b, ou bien:		HT	YOU HAVE C1 YOU HAVE C1C2 YOU HAVE cartes
5b	Vous êtes servi. Le calculateur montre sa main et tire ou passe, selon son jeu.		S	I HAVE cartes . . YOUR BANK IS \$ banque
6	Recommencez à l'instruction 4, si vous le désirez.			
*	<p>Note:</p> <ul style="list-style-type: none"> – Si vous faites blackjack à l'instruction 4, le message BLACKJACK est affiché, et (STAND) est exécuté automatiquement. – C représente une carte (n'importe laquelle), «cartes» est une chaîne de cartes. <p>Les numéros des cartes sont collés les uns aux autres, si bien qu'un 10 et un 7 apparaissent comme 107. K=roi, Q=dame, J=valet.</p>			

Exemple:

Battez le sabot, introduisez une racine de π et jouez au Blackjack avec mise de 2 \$.

<p>Appuyez sur</p> <p>[ASN] [ALPHA] DL [ALPHA] [Σ+] →</p> <p>[ASN] [ALPHA] HT [ALPHA] [1/x] →</p> <p>[ASN] [ALPHA] S [ALPHA] [√x] [USER] →</p>	<p>Affichage</p> <p>ASN DL 11</p> <p>ASN HT 12</p> <p>ASN S 13</p>	<p>La fonction DL (distribution) est affectée à la touche [Σ+].</p> <p>HT est affectée à la touche [1/x].</p> <p>S est affectée à la touche [√x].</p>
---	--	--

XEQ ALPHA SH ALPHA	→	SHUFFLING	
0 STO 21	→	104	Les 104 cartes sont battues. Seul FRC (π) est utilisé.
π STO 00			Distribution avec une mise de 2.
2 DL	→	DEALING	
		I SHOW 2	Je (HP-41C) retourne un 2.
		YOU HAVE 107	Vous avez un 10 et un 7.
S	→	I HAVE 2J	J'ai un 2 et un valet (J=Jack).
		I HAVE 2JK	J'ai un 2, un valet (J) et un roi (K=King).
		BUST	J'ai perdu.
		YOUR BANK IS \$2	Distribution et affichage banque 2 \$.
DL	→	I SHOW 6	Je retourne un 6.
		YOU HAVE A5	Vous avez un As et un 5.
HT	→	YOU HAVE A57	Vous avez un As, un 5 et un 7.
HT	→	YOU HAVE A575	Vous avez un As, un 5, un 7 et un 5.
S	→	I HAVE 6K	J'ai un 6 et un roi.
		I HAVE 6K8	J'ai un 6, un roi et un 8.
		BUST	J'ai perdu.
		YOUR BANK IS \$4	Votre banque est de 4 \$.

Astuce de programmation

Après le chargement de ce programme, vous disposez de 11 registres pour écrire un programme de blackjack utilisant des stratégies de jeu ou de mise simples. Le programme listé ci-dessous permet au programme blackjack d'afficher les contenus des registres et l'état des indicateurs binaires, pour déterminer s'il faut tirer ou passer. Si le programme perd, il double sa mise et peut gagner. En ajoutant encore d'autres modules mémoire au HP-41C, vous pouvez essayer d'introduire des stratégies de jeu encore plus complexes.

Le programme suivant nécessite une taille mémoire portée à 28.

60 Blackjack

01 ♦ LBL "PL"		18 XEQ "HT"	
02 2	Nouvelle mise.	19 GTO 00	
03 SF 22		20 ♦ LBL 01	
04 ♦ LBL 02		21 FS? 09	Si, non blackjack,
05 XEQ "DL"	Distribution des	22 XEQ "S"	alors passez.
06 ♦ LBL 00	cartes.	23 RCL 27	
07 RCL 24	Affiche le score.	24 RCL 21	
08 12		25 STO 27	Conserve la dernière
09 ENTER ↑	Ajustement de la	26 -	banque.
10 10	valeur de l'As, si pas	27 X<0?	Si vous gagnez,
11 FS? 07	d'As ajustement	28 GTO "PL"	faites une nouvelle
12 CLX	supprimé.	29 X=0?	mise. S'il n'y a pas
13 -		30 GTO 02	de gagnant, utilisez
14 X<=Y?	Si score ≤ 12	31 2	la mise précédente.
15 GTO 01	ou si blackjack	32 ST* 22	Si vous perdez,
16 FC? 09	alors passez	33 GTO 02	doublez la mise.
17 GTO 01	sinon tirez une carte.	34 END	

<pre> 01+LBL "CRD " 02 CLA 03 ASTO 19 04 1 05 STO 15 06 RCL 00 07 9821 08 * 09 .211327 10 + 11 FRC 12 STO 00 13 RCL 14 14 * 15 INT 16 1 17 + 18+LBL 02 19 RCL IND 15 20 X>Y? 21 GTO 03 22 - 23 ISG 15 24+LBL 99 25 GTO 02 26+LBL 03 27 DSE IND 15 28+LBL 99 29 DSE 14 30 12 31 RCL 14 32 X>Y? 33 GTO 04 34 XE0 "SH" 35+LBL 04 36 RCL 15 37 STO 16 38 10 39 X<=Y? 40 GTO 00 41 X<>Y 42 STO 16 43 1 44 X=Y? 45 GTO A 46 CLA </pre>	<p>Routine de distribution d'une carte.</p> <p>Générateur de nombres aléatoires.</p> <p>S'il ne reste plus que 12 cartes, battre le sabot.</p> <p>Stockage de la carte.</p>	<pre> 47 ARCL Y 48 GTO 01 49+LBL 00 50 STO 16 51 CLX 52 10 53 X=Y? 54 GTO "10" 55 1 56 + 57 X=Y? 58 GTO J 59 1 60 + 61 X=Y? 62 GTO "Q" 63 "K" 64 GTO 01 65+LBL A 66 "A" 67 CF 07 68 GTO 01 69+LBL "0" 70 "Q" 71 GTO 01 72+LBL J 73 "J" 74 GTO 01 75+LBL "10" 76 "10" 77+LBL 01 78 ASTO 19 79 RCL 16 80 RTN 81+LBL "SH" 82 "SHUFFLI NG" 83 AVIEW 84 1.013 85 ENTER↑ 86 8 87+LBL 14 88 STO IND Y 89 ISG Y 90 GTO 14 91 104 92 STO 14 93 CLD </pre>	<p>Stockage des carte alpha.</p> <p>Sous-programme de reconstruction du sabot.</p>
<p>R00=Nombre aléatoire R01=Les As R02=Les 2 R03=Les 3 R04=Les 4</p>	<p>R05=Les 5 R06=Les 6 R07=Les 7 R08=Les 8 R09=Les 9</p>		

<p>94 CF 00 95 CF 01 96 CF 02 97 CF 03 98 CF 04 99 RTN 100♦LBL "DL" 101 CF 09 102 SF 07 103 ABS 104 INT 105 FS?C 22 106 STO 22 107 RCL 22 108 STO 20 109 SF 06 110 CLA 111 ASTO 26 112 ASTO 25 113 XEQ "CRD" " 114 RCL 15 115 STO 17 116 XEQ "CRD" " 117 STO 23 118 CF 08 119 FS? 07 120 SF 08 121 CLA 122 ARCL 19 123 ARCL 25 124 ASTO 25 125 "I SHOW" " 126 ARCL 25 127 AVIEW 128 SF 07 129 0 130 STO 24 131 XEQ "CRD" " 132 XEQ "PH" 133 XEQ "CRD" " 134 XEQ "PH" 135 RCL 24 136 10</p>	<p>Blackjack. Pas d'As.</p> <p>Utilisation de la mise précédente ou stockage de la nouvelle mise.</p> <p>Distribution de la première carte du calculateur.</p> <p>Distribution de la seconde carte du calculateur.</p> <p>Conserve l'indicateur d'As du calculateur.</p> <p>Main du calculateur.</p> <p>Affiche la première carte du calculateur. Pas d'As.</p> <p>Distribution de la première carte du joueur. Pas d'impression.</p> <p>Impression activée.</p> <p>Distribution de la seconde carte du joueur.</p>	<p>137 FS? 07 138 CLX 139 + 140 21 141 X*Y? 142 SF 09 143 FS? 09 144 RTN 145 21.5 146 STO 24 147 1.5 148 ST* 20 149 "BLACKJACK" CK" 150 AVIEW 151♦LBL "S" 152 CF 06 153 FS? 07 154 GTO 05 155 11 156 RCL 24 157 X>Y? 158 GTO 05 159 10 160 ST+ 24 161♦LBL 05 162 CF 07 163 FS? 08 164 SF 07 165 RCL 17 166 STO 15 167 XEQ 04 168 XEQ "DH" 169 FS? 07 170 GTO 07 171 11 172 RCL 23 173 X*Y? 174 GTO 07 175 21.5 176 STO 23 177 "I HAVE BLACKJACK" 178 "HK" 179 AVIEW 180 GTO 07 181♦LBL 06</p>	<p>Affichage de la main du joueur.</p> <p>S'il n'y a pas blackjack, armez l'indicateur 9.</p> <p>Blackjack.</p> <p>Allez directement à «STAND».</p> <p>S'il n'y a pas blackjack, branchement au label 05.</p> <p>Réarme l'indicateur d'As du calculateur.</p> <p>Recherche de la carte détenue par le calculateur.</p> <p>Affichage de la main du calculateur.</p> <p>S'il n'y a pas d'As, branchement au label 07.</p>
---	---	---	---

R10=Les 10
R11=Les J (valets)
R12=Les Q (dames)
R13=Les K (rois)
R14=#Cartes restant dans le sabot

R15=Compteur
R16=Valeur de la carte en cours
R17=Carte cachée du calculateur
R18=Inutilisé
R19=Carte alpha en cours

182 XEQ "CRD"	Tirage de carte par le calculateur.	227 RCL 24	Test de dépassement (21).
183 XEQ "DH"	Tirage ou passe par le calculateur? Si le joueur dépasse 21, refaire les mises. Si le joueur fait blackjack, branchement. Si le calculateur a plus de 17 points, alors branchement. S'il n'a pas d'As, le calculateur tire une carte. S'il a un As avec un total de points compris entre 7 et 11, le calculateur tire une carte.	228 Z1.5	
184♦LBL 07		Addition de 10 pour un As.	229 X>Y?
185 FS? 06	230 RTH		
186 GT0 09	Vérifie si le calculateur n'a pas dépassé 21.	231 "BUST"	Affichage de la main du joueur.
187 FC? 09		232 AVIEW	
188 GT0 08	Vérification de match nul.	233 GT0 05	Affichage de la main du calculateur.
189 RCL 23		234♦LBL "DB"	
190 17	Armement de l'indicateur de dépassement si le joueur a perdu sa mise. Si le joueur perd, réduire sa banque.	235 "BUST"	Traitement du match nul.
191 X<=Y?		236 AVIEW	
192 GT0 08	Affichage de sa nouvelle banque. Le joueur tire une carte.	237 0	Etat: Size=028 Fix=00 CF=29 Indic 21 correspondant à Indic 55
193 FS? 07		238 RTH	
194 GT0 06	Distribution d'une autre carte.	239♦LBL "PH"	R20=Perte du joueur R21=Banque du joueur R22=Mise R23=Total des points du calculateur R24=Total des points du joueur R25=Inutilisé R26=Main du joueur R27=Main du calculateur
195 11		240 ST+ 24	
196 RCL 23	Affichage de la nouvelle main.	241 CLA	F06 Le joueur a perdu F07 Armé=pas d'As Désarmé=As F08 Armé=le calculateur n'a pas d'As Désarmé=le calculateur a un As F09 Armé=pas de blackjack Désarmé=blackjack F29 Désarmé pour supprimer la virgule décimale. F21 Doit correspondre à l'indicateur de présence de l'imprimante (F55).
197 X>Y?		242 ARCL 26	
198 GT0 06		243 ARCL 19	
199 7		244 ASTO 26	
200 X>Y?		245 "YOU HAVE	
201 GT0 06		246 ARCL 26	
202 10		247 AVIEW	
203 ST+ 23		248 RTH	
204♦LBL 08		249♦LBL "DH"	
205 Z1.5		250 ST+ 23	
206 RCL 23		251 CLA	
207 X>Y?		252 ARCL 25	
208 XEQ "DB"		253 ARCL 19	
209 RCL 24		254 ASTO 25	
210 -		255 "I HAVE	
211 X=0?		"	
212 XEQ "P"		256 ARCL 25	
213 X>0?		257 AVIEW	
214 SF 06		258 RTH	
215♦LBL 09		259♦LBL "P"	
216 RCL 20		260 "A PUSH"	
217 FS? 06		261 AVIEW	
218 CHS		262 ST* 20	
219 ST+ 21			
220 "YOUR BANK IS \$"			
221 ARCL 21			
222 AVIEW			
223 RTH			
224♦LBL "HT"			
225 XEQ "CRD"			
226 XEQ "PH"			



Hewlett-Packard France :

Société Anonyme au capital de 55.243.000 F. régie par les articles 118 à 150 de la loi sur les sociétés commerciales. RCS. Corbeil Essonnes B 709 805 030

Siège Social/Bureau de vente d'Orsay : Avenue des Tropiques, Z.I. de Courtabœuf, boîte postale n° 6, 91401 Orsay Cédex, tél. (1) 907 78 25

Bureau de vente d'Aix-en-Provence : Place Romée de Villeneuve, immeuble Le Ligourès, 13100 Aix-en-Provence, tél. (42) 59 41 02

Bureau de vente de Bordeaux : Avenue du Président-Kennedy, 33700 Mérignac, tél. (56) 97 01 81

Bureau de vente d'Evry : Tour Lorraine, boulevard de France, 91035 Evry Cédex, tél. (1) 077 96 60

Bureau de vente de Lille : Rue Van Gogh, immeuble Péricentre, 59650 Villeneuve d'Ascq, tél. (20) 91 41 25

Bureau de vente de Lyon : Chemin des Mouilles, boîte postale n° 162, 69130 Ecully Cédex, tél. (78) 33 81 25

Bureau de vente de Metz : 60, route de Metz, 57130 Jouy-aux-Arches, tél. (87) 69 45 32

Bureau de vente de Paris-Nord : Centre d'Affaires Paris-Nord, bâtiment Ampère, rue de la Commune de Paris, boîte postale n° 300, 93153 Le Blanc Mesnil Cédex, tél. (1) 931 88 50

Bureau de vente de Rennes : 2, allée de la Bourgonnette, 35100 Rennes, tél. (99) 51 42 44

Bureau de vente de Strasbourg : 18, rue du Canal de la Marne, 67300 Schiltigheim, tél. (88) 83 08 10

Bureau de vente de Toulouse : Péricentre de la Cèpière, 20, chemin de la Cèpière, 31081 Toulouse Cédex, tél. (61) 40 11 12

Hewlett-Packard Benelux S.A./N.V. :

Avenue du Col-Vert 1, B-1170 Brussels, tél. (02) 660 50 50

Hewlett-Packard (Schweiz) AG :

Château Bloc 19, CH-1219 Le Lignon-Genève, tél. (022) 96 03 22

Hewlett-Packard S.A., pour les pays du bassin méditerranéen, Afrique du Nord et Moyen-Orient :

35, Kolokotroni Street, Kifissia, GR-Athènes, tél. 80 81 741-4

Hewlett-Packard Ges.m.b.H., pour les pays socialistes :

Handelskai 52, boîte postale n° 7, A-1205 Vienne, tél. (0222) 35 16 21 à 27

Hewlett-Packard (Canada) Ltd. :

275, Hymus Boulevard, Pointe-Claire, Québec H9R1G7, tél. (514) 697-4232

Hewlett-Packard S.A., direction pour l'Europe :

7, rue du Bois-du-Lan, boîte postale, CH-1217 Meyrin 2, Genève, tél. (022) 82 70 00